

```

100 REMark PROGRAMA: COMUNICACIONES QDOS/MS-DOS v1.01
110 REMark AUTOR: Alejandro Alonso Puig
120 REMark FECHA: DIC-1987
130 REMark
140 POKE 163976,255
150 add = RESPR(1526): LBYTES MDV1_ram_bin, add: CALL add
160 BEEP 32000,1,200,5,2
170 WINDOW 512,256,0,0:PAPER 2:INK 7:MODE 4:BORDER 1,7
180 CSIZE 2,1:PRINT " COMUNICACIONES ";:CSIZE 0,1
190 PRINT "QDOS/MS-DOS v1.01 por Alejandro Alonso Puig 1987":CSIZE 0,0
200 WINDOW 512,212,0,22:PAPER 0:INK 7:BORDER 1,7
210 WINDOW#2,512,212,0,22:PAPER#2,0:BORDER#2,1,7
220 WINDOW#0,512,22,0,234:BORDER#0,1,7:PAPER#0,2
230 Baudios=1200:BAUD Baudios
240 OPEN#4,ser2
250 REPEAT programa
260 CLS#0:CSIZE#0,2,1
270 PRINT#0,"Velocidad transmision: ";Baudios;" baudios";
280 CSIZE#0,0,0
290 CLS
300 PRINT \\
310 PRINT " MENU PRINCIPAL"\\
320 PRINT " 1 .. Modo Terminal"
330 PRINT " 2 .. Transferencia fichero ASCII"
340 PRINT " 3 .. Cambiar velocidad de transmision"
350 PRINT " 4 .. Directorio"
360 PRINT " 5 .. Formatear"
370 PRINT " 6 .. Salir"
380 PRINT \\
390 INPUT " ELIGE OPCION: ";Opcion
400 SELECT ON Opcion
410 ON Opcion=1:TERMINAL
420 ON Opcion=2:TransFich
430 ON Opcion=3:SPEED
440 ON Opcion=4:Directorio
450 ON Opcion=5:Formatear
460 ON Opcion=6:EXIT programa
470 ON Opcion=REMAINDER :PRINT "OPCION INCORRECTA!"
480 END SELECT
490 CLS#0
500 CSIZE#0,1,1:INPUT#0,"PULSE RETURN PARA VOLVER AL MENU PRINCIPAL";Nada$
510 CSIZE#0,0,0
520 END REPEAT programa
530 CLOSE #4
540 CLS#0
550 STOP
1000 REMark
1010 REMark ##### PROCEDIMIENTO: TERMINAL
1020 REMark
1030 DEFINE PROCEDURE TERMINAL
1040 CLS:CLS#0:CSIZE#0,2,1
1050 PRINT#0," ^Q para finalizar";:CSIZE#0,0,0
1060 prompt$="A > "
1070 REPEAT envios
1080 PRINT prompt$;:INPUT orden$
1090 IF orden$=="^Q" THEN EXIT envios
1100 IF orden$=="^Z" THEN
1110 PRINT#4,CHR$(26);
1120 ELSE
1130 PRINT#4,orden$&CHR$(13)
1140 END IF
1150 IF orden$=="B:" THEN prompt$="B > "
1160 IF orden$=="A:" THEN prompt$="A > "
1170 END REPEAT envios
1180 END DEFINE TERMINAL
2000 REMark
2010 REMark ##### PROCEDIMIENTO: SPEED
2020 REMark

```



```

2030 DEFine PROCedure SPEED
2040   CLS
2050   PRINT \\
2060   PRINT " 1 .. 75 baudios"
2070   PRINT " 2 .. 300 baudios"
2080   PRINT " 3 .. 600 baudios"
2090   PRINT " 4 .. 1200 baudios"
2100   PRINT " 5 .. 2400 baudios"
2110   PRINT " 6 .. 4800 baudios"
2120   PRINT " 7 .. 9600 baudios"
2130   PRINT " 8 .. 19200 baudios (solo en transmisiones)"
2140   PRINT \\
2150   INPUT " ELIJA OPCION: ";Opcion
2160   SElect ON Opcion
2170     ON Opcion=1:Baudios=75
2180     ON Opcion=2:Baudios=300
2190     ON Opcion=3:Baudios=600
2200     ON Opcion=4:Baudios=1200
2210     ON Opcion=5:Baudios=2400
2220     ON Opcion=6:Baudios=4800
2230     ON Opcion=7:Baudios=9600
2240     ON Opcion=8:Baudios=19200
2250     ON Opcion=REMAINDER :PRINT "ERROR EN SELECCION DE VELOCIDAD!"
2260   END SElect
2270   BAUD Baudios
2280 END DEFine SPEED
3000 REMark
3010 REMark ##### PROCEDIMIENTO: TransFich
3020 REMark
3030 DEFine PROCedure TransFich
3040   CLS:PRINT \\
3050   PRINT " 1 .. Transferir de MS-DOS a QDOS"
3060   PRINT " 2 .. Transferir de QDOS a MS-DOS"
3070   PRINT \\
3080   INPUT " ELIGE OPCION: ";Opcion
3090   SElect ON Opcion
3100     ON Opcion=1:MSDOSaQDOS
3110     ON Opcion=2:QDOSaMSDOS
3120     ON Opcion=REMAINDER :PRINT "ERROR EN LA ELECCION!"
3130   END SElect
3140   DEFine PROCedure QDOSaMSDOS
3150     INPUT " Fichero a transferir: ";fich$
3160     OPEN#5,fich$
3170     PRINT " Transfiriendo ";
3180     REPEAT Transferencia
3190       IF EOF(#5) THEN EXIT Transferencia
3200       PRINT ".";
3210       INPUT#5,linea$
3220       PRINT#4,linea$;CHR$(13);REMark  A /ADIR CHR$(13)
3230     END REPEAT Transferencia
3240     PRINT#4,CHR$(26);
3250     CLOSE#5
3260     PRINT " transferido.":PAUSE 100
3270   END DEFine QDOSaMSDOS
3280   DEFine PROCedure MSDOSaQDOS
3290     INPUT " Fichero destino: ";fich$
3300     IF fich$(1 TO 4)=="RAM1" THEN
3310       OPEN_NEW#5,fich$
3320       Ram=1
3330     ELSE
3340       OPEN_NEW#5,ram1_tmporal
3350       Ram=0
3360     END IF
3370     PRINT " Recibiendo ";
3380     REPEAT QuitarMierda
3390       INPUT#4,linea$
3400       IF linea$=="TOF"&CHR$(13) THEN EXIT QuitarMierda
3410     END REPEAT QuitarMierda

```



```

3420 REPEAT recepcion
3430 PRINT ".";
3440 REPEAT MasMierda
3450 INPUT#4,linea$
3460 IF linea$=="TOF#"&CHR$(13):ELSE EXIT MasMierda
3470 END REPEAT MasMierda
3480 IF linea$=="EOF#"&CHR$(13) THEN EXIT recepcion
3490 PRINT#5,linea$(1 TO LEN(linea$)-1):REMark quitar chr$(13)
3500 END REPEAT recepcion
3510 CLOSE#5
3520 PRINT " recibido."
3530 IF Ram=0 THEN COPY ram1_tmporal TO fich$;DELETE Ram1-tmporal
3540 END DEFINE MSDOSaQDOS
3550 END DEFINE TransFich
4000 REMark
4010 REMark ##### PROCEDIMIENTO: Directorio
4020 REMark
4030 DEFINE PROCEDURE Directorio
4040 LOCAL fila,columna
4050 INPUT "Dispositivo? [ Ram1_ ] :";Dev$
4060 IF Dev$="" THEN Dev$="Ram1_"
4070 OPEN_NEW#7,ram1_Dir
4080 DIR#7,Dev$
4090 CLOSE#7
4100 CLS
4110 OPEN_IN#7,ram1_Dir
4120 INPUT#7,Volumen$
4130 INPUT#7,Sectores$
4140 CSIZE 2,1:PRINT Volumen$,Sectores$:CSIZE 0,0
4150 fila=3:columna=1
4160 REPEAT Presentar
4170 IF EOF(#7) THEN EXIT Presentar
4180 INPUT#7,Fichero$
4190 AT fila,columna:PRINT Fichero$
4200 IF fila<20 THEN
4210 fila=fila+1
4220 ELSE
4230 fila=3
4240 columna=columna+20
4250 END IF
4260 END REPEAT Presentar
4270 CLOSE#7:DELETE ram1_Dir
4280 END DEFINE Directorio
5000 REMark
5010 REMark ##### PROCEDIMIENTO: Formatear
5020 REMark
5030 DEFINE PROCEDURE Formatear
5040 CLS#0
5050 INPUT "Dispositivo a formatear? [ Ram1_ ]: ";Dev$
5060 IF Dev$="" THEN Dev$="Ram1_"
5070 PRINT "PULSE RETURN PARA CONFIRMAR FORMATEO DE ";Dev$:INPUT Nada$;
5080 PRINT "Formateando ... ";
5090 FORMAT Dev$
5100 PRINT " ... Formateado."
5110 END DEFINE Formatear

```